
Threat hunting introduction - why we need to hunt

Why context matters?

● Cyber threat hunting means focus on undetected threats in your 
infrastructure

● Finding additional pieces of attack could completely change the 
vision of ongoing threat

● Understanding the whole picture of attack allows you to pay 
attention to specific parts of your defending bastion (patch 
management, checking certain logs, protect mobile, etc)

● Learning the actor TTPs and malware attribution lead to 
improvement of your threat model 

Threat hunting introduction - Yara in a nutshell

● “Superb version” of grep tool

● Industry standard for sharing knowledge

● Main purpose:

○ Malware detection and classification

○ Finding suspicious activity

○ Searching for intellectual property

○ Forensic analysis

● Lots of useful documentation here

https://yara.readthedocs.io/en/stable/

Threat hunting introduction - Yara in a nutshell

Yara on malware samples

Exact detection

● AV-like detection to detect specific IOCs

● To directly rise an alert and pay max attention

● Very few false positives

Threat hunting introduction - Yara in a nutshell

● For example, this Yara rule is detecting only specific ID, which is not directly related to Rubeus workflow and 
could be easily changed.

● There is also a magic number filter (0x5A4D) to get only MZ files.

rule HackTool_MSIL_Rubeus_1 {

    meta:

        description = "The TypeLibGUID present in a .NET binary maps directly to the ProjectGuid found in the '.csproj' file of a .NET 

project. This rule looks for .NET PE files that contain the ProjectGuid found in the public Rubeus project."

        md5 = "66e0681a500c726ed52e5ea9423d2654"

        rev = 4

        author = "FireEye"

    strings:

        $typelibguid = "658C8B7F-3664-4A95-9572-A3E5871DFC06" ascii nocase wide

    condition:

        uint16(0) == 0x5A4D and $typelibguid

}

https://isc.sans.edu/forums/diary/Writing+Yara+Rules+for+Fun+and+Profit+Notes+from+the+FireEye+Breach+Countermeasures/26870/

Threat hunting introduction - Yara in a nutshell

Yara on malware samples

Exact detection

● AV-like detection to detect specific IOCs

● To directly rise an alert and pay max attention

● Very few false positives

Hunting detection

● Using more generic approach to find not only 
samples from the initial attack, but also all 
related IOCs (other campaigns, historical)

● Allows to uncover additional actor’s activities

● Perfectly suited for getting a feed of samples to 
filter manually

● Could get lots of false positives

Threat hunting introduction - Yara in a nutshell

● In this example, we are using “any of them” condition to find files contained at least one of specified entities.

● There is no any additional conditions like file-type filtering.

rule hijack_network {

    meta:

        author = "x0r"

        description = "Hijack network configuration"

  version = "0.1"

    strings:

        $p1 = "SOFTWARE\\Classes\\PROTOCOLS\\Handler" nocase

        $p2 = "SOFTWARE\\Classes\\PROTOCOLS\\Filter" nocase

        $p3 = "Microsoft\\Windows\\CurrentVersion\\Internet Settings\\ProxyServer" nocase

        $p4 = "software\\microsoft\\windows\\currentversion\\internet settings\\proxyenable" nocase

        $f1 = "drivers\\etc\\hosts" nocase

    condition:

        any of them

}

https://github.com/Yara-Rules/rules/blob/master/capabilities/capabilities.yar

Yara services at VirusTotal

● Searching in the past - Retrohunt

○ Scans the dataset of previously submitted files (up to 1 year) - more than 
500M files (~680TB worth of data) in 2-3 hours and reports you the files 
that matches your rules

○ Perfect tool for attribution, allows you to uncover additional historical 
activity of the same actor

○ Also could be used to quickly check the rule for false positives before 
submitting to production (less than 1min)

Yara services at VirusTotal - Retrohunt

Unknown Android spyware sample

Retrohunt

Historical X-Agent Windows samples used by  Sofacy

https://www.virustotal.com/gui/file/c8ccf898705bd4f691aed68a208710785af341d854043d2e3644fdd1832c356f

Yara services at VirusTotal

● Searching in the future - Livehunt

○ Daily scans of all incoming samples

○ Suitable for tracking ongoing campaigns/further activities

○ Could be also used to track leaked data, brand protection, etc

● Searching in the past - Retrohunt

○ Scans the dataset of previously submitted files (up to 1 year) - more than 
500M files (~680TB worth of data) in 2-3 hours and reports you the files 
that matches your rules

○ Perfect tool for attribution, allows you to uncover additional historical 
activity of the same actor

○ Also could be used to quickly check the rule for false positives before 
submitting to production (less than 1min)

Yara services at VirusTotal - Livehunt

***
WATERMARK

THIS FILE IS PROPERTY OF CORPNAME. 
DO NOT UPLOAD IT TO THE MULTISCANNER 

SERVICES OR ANY OTHER 3RD PARTY 
***

Rule on Livehunt Notification alert on match

Yara VirusTotal module

● This Yara module created specially for Livehunt service to operate with all 
available VT context data

● Yara is a static analyzer by default, however with this module you can avoid 
static obfuscation making signatures on a malware behaviour executed in our 
sandboxes

● We have a number of third-parties SB (Vmray, Lastline, Tencent, C2AE, etc), as 
well as our own  Jujubox

● Provides features in different categories:

○ Metadata (AVs rate, ExifTool, submissions, type, signature, VT tags, etc) 

○ Behaviour (network, file system, SB verdict, Android/Windows specific)

● Our documentation is here 

https://blog.virustotal.com/2019/10/in-house-dynamic-analysis-virustotal-jujubox.html
https://support.virustotal.com/hc/en-us/articles/360007088057-Writing-YARA-rules-for-Livehunt

Yara VirusTotal module - Sofacy behaviour

Check out Sofacy’s sample from 2017

import "vt"

rule Sofacy_2018 {
condition:

for any mutex in vt.behaviour.mutexes_created : (
       mutex == "OEyZrUgaebnWBiSFZBZreuicRMbbcqHAXkqzLMAJIitMO"
    )
}

https://www.virustotal.com/gui/file/dea3a99388e9c962de9ea1008ff35bc2dc66f67a911451e7b501183e360bb95e/behavior

Yara VirusTotal module - Metadata exploring 

● vt.metadata.analysis_stats.malicious - number of antivirus engines that detected the file as malicious.

● vt.metadata.analysis_stats.failure - number of antivirus engines that failed scanning the file.

● vt.metadata.analysis_stats.type_unsupported - number of antivirus engines that don't support the file's type.

● vt.metadata.submitter.country - country from where the file was submitted. Two-letter ISO 3166, in uppercase.

● vt.metadata.unique_sources - number of unique sources that have submitted this file.

● vt.metadata.times_submitted - number of times the file has been submitted to VirusTotal.

● vt.metadata.new_file - True if the file has been submitted to the VirusTotal for the first time.

import "vt"

rule Anomalies_test {    
condition:

vt.metadata.analysis_stats.malicious > 5 and
vt.metadata.submitter.country == “CN” and
vt.metadata.unique_sources == 1 and
vt.metadata.times_submitted == 1 and
vt.metadata.new_file and
(vt.metadata.analysis_stats.failure > 10 or vt.metadata.analysis_stats.type_unsupported > 20)

}

Yara VirusTotal module

import "vt"

rule Colibri_Loader {
meta:

author = "Virustotal"
created = "2022.04.26"
refer = 

"https://blog.malwarebytes.com/threat-intelligence/2022/04/colibri-loader-combines-t
ask-scheduler-and-powershell-in-clever-persistence-technique/"

hash = "7c011de51c6d2a058d82101d314bc55b2299d89e"
hash = "1e1af9c4475419b5c73e57a8a9938ef840e2652c"

    
condition:

for any c in vt.behaviour.http_conversations: (
c.url contains "/securetunnel.co:" or c.url contains "/securetunnel.co/" or      

c.url contains "/trkal0.dot" or c.url contains "/vpnchecker.php"
            )

        or

        for any file_dropped in vt.behaviour.files_dropped: (
file_dropped.path contains "\\vpnchecker[1].htm"
)

or
      (
      for any file_dropped in vt.behaviour.files_dropped: (

file_dropped.path contains "\\WIKWAFRE\\install[1].exe"
)

      and 
      vt.metadata.analysis_stats.malicious > 10
      )
}

https://blog.malwarebytes.com/threat-intelligence/2022/04/colibri-loader-combines-task-scheduler-and-powershell-in-clever-persistence-technique/
https://blog.malwarebytes.com/threat-intelligence/2022/04/colibri-loader-combines-task-scheduler-and-powershell-in-clever-persistence-technique/
https://www.virustotal.com/gui/file/666268641a7db3b600a143fff00a063e77066ad72ac659ebc77bb5d1acd5633d/behavior/C2AE
https://www.virustotal.com/gui/file/b92f4b4684951ff2e5abdb1280e6bff80a14b83f25e4f3de39985f188d0f3aad/behavior

Yara VirusTotal module

Yara VirusTotal module

chat.zaqxswcdevfrbgtnhymjukilop[.]online

https://www.virustotal.com/gui/search/behaviour_network%253Achat.zaqxswcdevfrbgtnhymjukilop.online

Yara VirusTotal module

chat.zaqxswcdevfrbgtnhymjukilop[.]online

import "vt"

rule OrigamiElephant_c2 {
meta:

author = "Virustotal"
created = "2022.04.27"
hash = "949233b8691680db9826b1a9a21926737149f5ae"

    
condition:

for any d in vt.behaviour.dns_lookups: (
d.hostname contains "chat.zaqxswcdevfrbgtnhymjukilop.online")

        or
for any t in vt.behaviour.ip_traffic: (

t.destination_ip contains "131.153.22.218")
  or

for any text in vt.behaviour.text_decoded: (
text == "chat.zaqxswcdevfrbgtnhymjukilop.online" or
text == "this is agdfhgdfh key"

            )
}

https://www.virustotal.com/gui/search/behaviour_network%253Achat.zaqxswcdevfrbgtnhymjukilop.online
https://www.virustotal.com/gui/file/49ede2937a565ffe13f1212c8c67a8a7828b4ce7ede51b7753d597ec21855d6e/details

Yara VirusTotal module

Other Yara modules

● PE - operate with attributes and features of PE file format (header fields)
○ pe.timestamp >= 1424563200

● Dotnet - similar to PE but for .NET files
○ dotnet.assembly.name == "Keylogger"

● ELF - similar to PE but for ELF files
○ elf.number_of_sections == 1

● Cuckoo - integrates with Cuckoo sandbox, allows retrieving behavioural info
○ cuckoo.network.host(/192\.168\.1\.1/)

● Magic - simplifies magic byte checking (Linux “file” tool) 
○ magic.type() contains "PDF"

● Hash - allows you to calculate hashes from portions of file/string 
○ hash.md5(0, filesize) == "feba6c919e3797e7778e8f2e85fa033d"

● Math - allows you to calculate certain values from portions of your file 
○ math.entropy(0, filesize) >= 7 

● Time - epoch time, useful to implement some dynamic time-based condition
○ pe.timestamp > time.now()

rule ZIP_Header {    
condition:

uint16(0) == 0x4b50
}

import "magic"

rule ZIP_Header {    
condition:

magic.type() contains "Zip archive"
} 

https://yara.readthedocs.io/en/v4.2.1/modules/pe.html
https://yara.readthedocs.io/en/v4.2.1/modules/dotnet.html
https://yara.readthedocs.io/en/v4.2.1/modules/elf.html
https://yara.readthedocs.io/en/v4.2.1/modules/cuckoo.html
https://yara.readthedocs.io/en/v4.2.1/modules/magic.html
https://yara.readthedocs.io/en/v4.2.1/modules/hash.html
https://yara.readthedocs.io/en/v4.2.1/modules/math.html
https://yara.readthedocs.io/en/v4.2.1/modules/time.html

PE module

import "pe"

rule Sednit_XTunnel {
meta:

author = "Virustotal"
description = "Sednit attack on Bundestag in 2015"
refer = 

"https://www.welivesecurity.com/wp-content/uploads/2016/10/eset-sednit-part-2.pdf"

condition:
pe.pdb_path contains "XAPS_OBJECTIVE"

}

https://www.welivesecurity.com/wp-content/uploads/2016/10/eset-sednit-part-2.pdf

PE module

import "pe"

rule SUSP_NVIDIA_LAPSUS_Leak_Compromised_Cert_Mar22_1 {
   meta:
      description = "Detects a binary signed with the leaked NVIDIA certifcate and 
compiled after March 1st 2022"
      author = "Florian Roth"
      date = "2022-03-03"
      modified = "2022-03-04"
      score = 70
      reference = "https://twitter.com/cyb3rops/status/1499514240008437762"
   condition:

      uint16(0) == 0x5a4d and filesize < 100MB and

      pe.timestamp > 1646092800 and  // 2022-03-01, comment out to find all files signed 
with that certificate

      for any i in (0 .. pe.number_of_signatures) : (
         pe.signatures[i].issuer contains "VeriSign Class 3 Code Signing 2010 CA" and 

(
         pe.signatures[i].serial == "43:bb:43:7d:60:98:66:28:6d:d8:39:e1:d0:03:09:f5"

or
         pe.signatures[i].serial == "14:78:1b:c8:62:e8:dc:50:3a:55:93:46:f5:dc:c5:18"
         )
     )

}

PE module

import "pe"

rule Suspicious_too_old_x64_PE {
    meta:
        author = "Kaspersky"
        description = "Rule based on TripleFantasy from Equation group"

    condition:
    (uint16(0) == 0x5A4D) 

  and (pe.machine == pe.MACHINE_AMD64 or pe.machine == pe.MACHINE_IA64)
  and pe.timestamp > 631155661 // 1990-01-01
  and pe.timestamp < 1072915200 // 2004-01-01
  and filesize < 2000000

}

VirusTotal Diff - what are Yara generators?

One-click tool to make Yara ruleset

● Pros:

○ Fast!

○ Quite useful to make exact detections

● Cons:

○ Need to deploy and maintain ‘clean’ collection

○ Rule quality highly depends on whitelisting quality 
(whitelisting ~ FP number)

○ Could be overfitting with 0 additional detects

VirusTotal Diff - [DEMO]

● Our own binary diff tool weaponized with signature generator

● Most useful while dealing with a bunch of similar samples 
(same malware family) to make generic signatures

● Most suitable for widely spreaded malware 

● Also could be used as an effective RE tool to highlight common 
and unique code instead of reversing it manually in decompiler

● Provides one-click rules for Livehunt and Retrohunt

● We can instantly pivot on specific signatures to get more 
samples or to check for uniqueness  

● Let’s explore the collection of dex files from PhantomLance 
campaign conducted by OceanLotus group

https://support.virustotal.com/hc/en-us/articles/360010904818-VT-DIFF-Automatic-YARA-rules
https://www.virustotal.com/gui/collection/1a6e248e92a5ff9c5756a71f4500669b1602b42756d32ccb9282fd70f6cc578f
https://securelist.com/apt-phantomlance/96772/
https://securelist.com/apt-phantomlance/96772/

—
Thank you

Alexey Firsh
@alexey_firsh

brighttalk.com/webcast/18282/573479

https://www.brighttalk.com/webcast/18282/573479

